


Curso de

Lógica de Programação



Por

Jeiverson C. V. M. Santos

Aula 13

Procedimentos

"Rotina não é fazer sempre a mesma coisa, mas fazer a mesma coisa sempre do mesmo jeito."



Joel de Sousa



Primeiro

Vamos ver o resultado do "para casa"




e do desafio



Procedimento

- Procedimento é uma rotina.
- É um pedaço do código que vai se repetir.



Exemplo

Procedimento para se trocar as cordas do violão.


- Pegue uma corda nova
 - Retire uma corda velha
 - coloque a corda nova no lugar da velha
 - amarre as pontas da corda
 - aperte a tarracha para afinar
-
- Pegue uma corda nova
 - Retire uma corda velha
 - coloque a corda nova no lugar da velha
 - amarre as pontas da corda
 - aperte a tarracha para afinar

•
•
•



Ao invés de escrever 6 vezes aquilo tudo...

- TrocarCorda()
- TrocarCorda()
- TrocarCorda()
- TrocarCorda()
- TrocarCorda()
- TrocarCorda()



Por que não usar um comando de repetição?

TrocarCorda()


</>



</>

TrocarCorda()

</>



</>

TrocarCorda()

</>



</>

</>



</>

TrocarCorda()

TrocarCorda()

TrocarCorda()



Porque eu posso chamar o procedimento a qualquer hora.

Exemplo - sem Procedimento

```
1 algoritmo "Detector do maior"
2 var
3     n, m, i: inteiro
4     // n: número | m: maior | i: contador
5 inicio
6     m <- 0
7     escreval("-----")
8     escreval("Detector do maior")
9     escreval("Maior até agora:", m)
10    escreval("-----")
11    para i <- 1 ate 5 faca
12        escreva("Digite um número: ")
13        leia(n)
14        se n > m entao
15            m <- n
16        fimse
17        limpatela
18        escreval("-----")
19        escreval("Detector do maior")
20        escreval("Maior até agora:", m)
21        escreval("-----")
22    fimpara
23 fimalgoritmo
```


Exemplo - com procedimento

```
1 algoritmo "Detector do maior"
2 var
3   n, m, i: inteiro
4   // n: número | m: maior | i: contador
5 procedimento EscreveTexto()
6 inicio
7   limpatela
8   escreval("-----")
9   escreval("Detector do maior")
10  escreval("Maior até agora:", m)
11  escreval("-----")
12 fimprocedimento
13 inicio
14   m <- 0
15   EscreveTexto()
16   para i <- 1 ate 5 faca
17     escreva("Digite um número: ")
18     leia(n)
19     se n > m entao
20       m <- n
21     fimse
22   EscreveTexto()
23   fimpara
24 fimalgoritmo |
```


Não diminuiu, mas...

- Ficou mais fácil de ler.
- Ficou mais organizado.
- Ficou mais fácil de corrigir.



Passagem de Parâmetro

Por valor




Exemplo



```
procedimento soma(a, b: inteiro)  
inicio  
    escreva(a, " + ", b, " = ", a+b)  
fimprocedimento
```



Exemplo





```
procedimento soma(a, b: inteiro)  
var  
    resultado: inteiro  
inicio  
    resultado <- a + b  
    escreva(a, " +", b, " =", resultado)  
fimprocedimento
```


Exemplo

```
1 algoritmo "Usando Procedimento"
2 var
3 procedimento soma(a, b: inteiro)
4 var
5     resultado: inteiro
6 inicio
7     resultado <- a + b
8     escreva(a, " +", b, " =", resultado)
9 fimprocedimento
10 inicio
11     soma(2, 3)
12 fimalgoritmo
```




Exemplo



```
1 algoritmo "Usando Procedimento"
2 var
3   x, y: inteiro
4 procedimento soma(a, b: inteiro)
5 var
6   resultado: inteiro
7 inicio
8   resultado <- a + b
9   escreva(a, " +", b, " =", resultado)
10 fimprocedimento
11 inicio
12   x <- 2
13   y <- 3
14   soma(x, y)
15 fimalgoritmo |
```


Botão Passo a Passo




Escopo das variáveis

Áreas das variáveis de memória (Globais e Locais)			
Escopo	Nome	Tipo	Valor



Exercício

Fazer um algoritmo que, usando procedimento, verifique se um número é par ou ímpar.



Resposta

```
1 algoritmo "Par | Ímpar"
2 var
3     n: inteiro
4 procedimento verifica(n: inteiro)
5 inicio
6     se n % 2 = 0 entao
7         escreva (n, " é par.")
8     senao
9         escreva(n, " é ímpar.")
10    fimse
11 fimprocedimento
12 inicio
13     escreva("Digite um número: ")
14     leia(n)
15     verifica(n)
16 fimalgoritmo |
```



Teste de Escopo


```
1 algoritmo "Teste de escopo"
2 var
3   n: inteiro
4 procedimento rotina(x: inteiro)
5 var
6   v: inteiro
7 inicio
8   v <- x + 3
9   escreva(v)
10 fimprocedimento
11 inicio
12   n <- 2
13   escreval(n)
14   rotina(n)
15   escreva(v)
16 fimalgoritmo |
```

v, x: local

n: global

Passagem de Parâmetro

Por referência



O que será escrito?

```
1 algoritmo "Parâmetro por referência"
2 var
3     n: inteiro
4 procedimento rotina(var x: inteiro)
5 inicio
6     x <- x + 3
7 fimprocedimento
8 inicio
9     n <- 2
10    escreval(n)
11    rotina(n)
12    escreval(n)|
13 fimalgoritmo
```

The diagram illustrates the flow of data and variable references. It shows how the variable `n` is passed by reference from the main block to the `rotina` procedure. The arrows indicate that the `var` keyword in the procedure signature and the `var` keyword in the procedure call are linked, as are the `n` parameters and the `n` variable in the main block.

Para Casa

1. Fazer um algoritmo que, usando procedimento com passagem de parâmetro por valor, escreva quantos termos o usuário quiser da sequência: 1, 3, 5, 7...
2. Fazer um algoritmo que, usando procedimento com passagem de parâmetro por referência, calcule a raiz quadrada de um número. Escreva o valor da raiz quadrada no programa principal, não na função.

Dicas: Para calcular a raiz de um número, basta elevar esse número a $1/2$. Uma raiz quadrada é um número Real.

